

实验12: **ROM + RAM**



1

ROM

ROMs

方法1：数据直接写到代码中

```
1 module ROM_asynRead(  
2     input logic [2:0] addr,  
3     output logic [7:0] data );  
4  
5  
6  
7  
8     always_comb // image ROM  
9         case (addr)  
10             3'h0: data = 8'b0011_1100; // ****  
11             3'h1: data = 8'b0111_1110; // *****  
12             3'h2: data = 8'b1111_1111; // *****  
13             3'h3: data = 8'b1111_1111; // *****  
14             3'h4: data = 8'b1111_1111; // *****  
15             3'h5: data = 8'b1111_1111; // *****  
16             3'h6: data = 8'b0111_1110; // *****  
17             3'h7: data = 8'b0011_1100; // ****  
18         endcase  
19  
20     logic rom_bit; // ROM中每个像素值  
21     assign rom_bit = data[3]; // ROM addr行中第3列值  
22 endmodule
```


异步

```
1 module ROM_synRead( input logic clk,  
2                     input logic [2:0] addr,  
3                     output logic [7:0] data );  
4  
5     logic [2:0] addr_reg;  
6     always_ff @(posedge clk)  
7         addr_reg <= addr;  
8  
9     always_comb // image ROM  
10        case (addr_reg)  
11            3'h0: data = 8'b0011_1100; // ****  
12            3'h1: data = 8'b0111_1110; // *****  
13            3'h2: data = 8'b1111_1111; // *****  
14            3'h3: data = 8'b1111_1111; // *****  
15            3'h4: data = 8'b1111_1111; // *****  
16            3'h5: data = 8'b1111_1111; // *****  
17            3'h6: data = 8'b0111_1110; // *****  
18            3'h7: data = 8'b0011_1100; // ****  
19        endcase  
20  
21     logic rom_bit; // ROM中每个像素值  
22     assign rom_bit = data[3]; // ROM addr行中第3列值  
23 endmodule
```

同步


方法2：初始化时赋值

```
1 // 4x8bit ROM
2 module ROM2(
3     input logic [1:0] addr,      // 2^2 = 4
4     output logic [7:0] oneWord );
5
6     parameter data = 32'h00_C8_F9_AF; //left index
7     parameter N_BITS = 8; //no. of bits in rom word
8     parameter N_WORDS = 4; //no. of words in rom
9     parameter TOTAL = N_BITS * N_WORDS - 1;
10
11     //      位宽      字宽
12     logic [N_BITS-1 : 0] rom [0 : N_WORDS-1];
13
14     integer i;
15     initial
16         for(i=0; i<N_WORDS; i=i+1)
17             rom[i] = data[(TOTAL - N_BITS * i) -: N_BITS];
18     //若i=1, 则 rom[1] = data[( 31 - 8 * 1) -: 8]=data[23 -: 8]=data[23:16]
19
20     assign oneWord = rom[addr];
21 endmodule
```



方法3：从文件中读取数据

如，从指令存储器**ROM**中取出指令



Verilog有2个系统任务，从文件中读取数据到存储器

`$readmemb`("<数据文件名>", <存储器名>, <起始地址>, <终止地址>);

`$readmemh`("<数据文件名>", <存储器名>, <起始地址>, <终止地址>);

```
1 // ROM: 64x32bit
2 module iMemory(input logic [5:0] addr ,
3                 output logic [31:0] data );
4
5 //      位宽      字宽
6 logic [31:0] ROM [63:0];    26 = 64
7
8 initial
9     $readmemh("Test.dat", ROM);
10
11 assign data = ROM[addr];
12 endmodule
```

Test.dat

1	20020005
2	2003000C
3	2067FFF7
4	00E22025
5	00642824
6	00A42820
7	10A7000A
8	0064202A
9	10800001
10	20050000
11	00E2202A
12	00853820
13	00E23822
14	AC670044
15	8C020050
16	08000011
17	20020001
18	AC020054

【参考】教材P276-278

方法4: IP核

Flow Navigator

PROJECT MANAGER

- Settings
- Add Sources
- Language Templates
- IP Catalog

单击IP目录

IP Catalog

Cores | Interfaces

Search: memory (3 matches)

Name: AXI4 Status

Memories & Storage Elements

- Memory Interface Generators
 - RAMs & ROMs
 - Distributed Memory Generator
 - RAMs & ROMs & BRAM
 - Block Memory Generator

AXI4 Production

Customize IP

Distributed Memory Generator (8.0)

Documentation IP Location Switch to Defaults

Show disabled ports

Component Name: iRom

memory config Port config RST & Initialization

Options

Depth: 64 [16 - 65536]

Data Width: 32 [1 - 1024]

Memory Type


Memory Type

- ☒ ROM
- ☐ Single Port RAM
- ☐ Simple Dual Port RAM
- ☐ Dual Port RAM


OK Cancel

6 / 16


用 .coe 文件初始化ROM




配置输入/输出引脚
是否具有锁存功能



仿真



```
1 `timescale 1ns / 1ps
2 module iROM_Sim( );
3     logic clk;
4     logic [5:0] a;
5     logic [31:0] spo;
6
7     //实例化
8     iRom imem (
9         .a(a),           // input wire [5 : 0] a
10        .clk(clk),        // input wire clk
11        .spo(spo));       // output wire [31 : 0] spo
12
13     initial begin
14         a = 0;    clk = 0;
15     end
16
17     always #10 clk = ~clk;
18
19     initial begin
20         #40 a = 1;
21         #30 a = 2;
22     end
23 endmodule
```



Nexys4 上的存储空间

- Artix-7 芯片有两种存储空间 (embedded memory)

① **Distributed RAM** 1,188Kb

由多个逻辑单元中的LUTs构成，故容量小

② **Block RAM** 4,860Kb

- The Nexys4 DDR board contains two external memories:

① *1Gib* (128MiB) DDR2 **SDRAM**,

② *128Mib* (16MiB) non-volatile(非易失) serial **Flash** device.

- 以10为底的指数: $1\text{KB}=10^3=1000$, $1\text{MB}=10^6=1000\text{KB}$

- 以 2为底的指数: $1\text{KiB}=2^{10}=1024$, $1\text{MiB}=2^{20}=1024\text{KiB}$




2

RAM

RAMs

单口 RAM：1套地址总线，1套数据总线，读、写分开

```
1 module singlePort_RAM_asynRead
2     #( parameter ADDR_WIDTH = 12, DATA_WIDTH = 8 )
3     (input logic clk,
4      input logic we,          // Write enable
5      input logic [ADDR_WIDTH-1 : 0] addr,
6      input logic [DATA_WIDTH-1 : 0] din,
7      output logic [DATA_WIDTH-1 : 0] dout );
8
9     //      位宽      字宽
10    logic [DATA_WIDTH-1 : 0] RAM [2**ADDR_WIDTH-1 : 0];
11
12    always_ff @(posedge clk)
13        if (we) RAM[addr] <= din; // write operation
14
15    assign      dout = RAM[addr]; // read operation 异步
16 endmodule
```



```
1 module singlePort_RAM_synRead
2     #( parameter ADDR_WIDTH = 12, DATA_WIDTH = 8 )
3     (input logic clk,
4      input logic we,          // Write enable
5      input logic [ADDR_WIDTH-1 : 0] addr,
6      input logic [DATA_WIDTH-1 : 0] din,
7      output logic [DATA_WIDTH-1 : 0] dout );
8
9     //      位宽      字宽
10    logic [DATA_WIDTH-1 : 0] RAM [2**ADDR_WIDTH-1 : 0];
11    logic [ADDR_WIDTH-1 : 0] addr_reg ;
12
13    always_ff @(posedge clk)
14        begin //=== 先写后读 ===
15            if (we) RAM[addr] <= din; // write operation
16            addr_reg <= addr;          //同步地址
17        end
18
19    assign      dout = RAM[addr_reg]; // read operation 同步
20 endmodule
```


Due to the internal structure, an asynchronous read operation can be realized **only** by the distributed RAM.

Simple Dual-port RAM: 2套地址总线, 1套数据总线

```
1 // read 和 write是并行的
2 module SimpleDualPortRAM1(
3     input logic          clk, wr_en,
4     input logic [ 9:0] rd_adr,
5     input logic [ 9:0] wr_adr,
6     input logic [15:0] data_in,
7     output logic [15:0] data_out );
8
9 //      位宽      字宽
10 logic [15:0] RAM [1023:0];
11
12 always_ff @( posedge clk )
13 begin
14     data_out <= RAM[rd_adr];           //Read RAM
15     if (wr_en) RAM[wr_adr] <= data_in; //Write RAM
16 end
17 endmodule
```

读、写并行：先保存原值，再写入


RAM中原有的值先存入data_out_reg中输出，
然后 data_in 再保存到RAM中。



并行：14、15行可以颠倒。

Dual-port RAM: 2套地址总线, 1.5套数据总线

```
1 module DualPort_RAM_asynRead
2     #(parameter ADDR_WIDTH = 6, DATA_WIDTH = 8 )
3     ( input  logic clk,
4       input  logic we,    // Write enable
5       input  logic [ADDR_WIDTH-1: 0] addr_a, addr_b ,
6       input  logic [DATA_WIDTH-1: 0] din_a,
7       output logic [DATA_WIDTH-1: 0] dout_a, dout_b );
8
9     //          位宽          字宽
10    logic [DATA_WIDTH-1: 0] RAM [2**ADDR_WIDTH-1: 0];
11
12    always_ff @(posedge clk)
13        if (we) RAM[addr_a] <= din_a; // write operation
14
15    assign dout_a = RAM[addr_a];    // read operations 1
16    assign dout_b = RAM[addr_b];    // read operations 2
17 endmodule
```




```
1 module DualPort_RAM_synRead
2     #(parameter ADDR_WIDTH = 6, DATA_WIDTH = 8 )
3     ( input  logic clk,
4       input  logic we,    // Write enable
5       input  logic [ADDR_WIDTH-1: 0] addr_a, addr_b,
6       input  logic [DATA_WIDTH-1: 0] din_a,
7       output logic [DATA_WIDTH-1: 0] dout_a, dout_b );
8
9     //          位宽          字宽
10    logic [DATA_WIDTH-1: 0] RAM [2**ADDR_WIDTH-1: 0];
11    logic [ADDR_WIDTH-1: 0] addr_a_reg, addr_b_reg;
12
13    always_ff @(posedge clk)
14    begin
15        if (we) RAM[addr_a] <= din_a; // write operation
16        addr_a_reg <= addr_a;
17        addr_b_reg <= addr_b;
18    end
19
20    assign dout_a = RAM[addr_a_reg]; // read operations 1
21    assign dout_b = RAM[addr_b_reg]; // read operations 2
22 endmodule
```

两个口都可以读写, 但不能对同一个地址同时写。

Can be realized only by distributed RAM, thus its size is limited.

Full Dual-port RAM: 地址线、数据线、时钟都2套

```
1  module FullDualPortMemory(  
2      input logic      clk_a, clk_b,  
3      input logic      wr_a,  wr_b,  
4      input logic [ 9:0] address_a, address_b,  
5      input logic [15:0] data_in_a, data_in_b,  
6      output logic[15:0] dat_out_a, dat_out_b );  
7  
8      logic [15:0] RAM [1023:0];  
9  
10     // Port a  
11     always_ff @( posedge clk_a )  
12     begin  
13         dat_out_a      <= RAM[address_a];  
14         if (wr_a)  
15         begin  
16             dat_out_a    <= data_in_a;  
17             RAM[address_a] <= data_in_a;  
18         end  
19     end
```



```
21     // Port b  
22     always_ff @( posedge clk_b )  
23     begin  
24         dat_out_b      <= RAM[address_b];  
25         if (wr_b)  
26         begin  
27             dat_out_b    <= data_in_b;  
28             RAM[address_b] <= data_in_b;  
29         end  
30     end  
31 endmodule
```

先输出
再写入

IP核设计RAM

The image displays the IP Catalog and two instances of the Customized Distributed Memory Generator (8.0) configuration window.


IP Catalog: The leftmost window shows the IP Catalog with the search term "ram". The "Distributed Memory Generator" is selected under the "RAMs & ROMs" category.

Customize IP - Distributed Memory Generator (8.0): This window shows the configuration for the Distributed Memory Generator (8.0). The Component Name is "dRAM". The "memory config" tab is active, showing the "Options" section with "Depth" set to 64 and "Data Width" set to 32. The "Memory Type" section shows "Single Port RAM" selected. The "Port config" tab is also visible, showing the "Input Options" section with "Non Registered" selected. The "Dual Port Address" section shows "Non Registered" selected. The "Output Options" section shows "Non Registered" selected. The "RST & Initialization" tab is also visible.


Diagram: A block diagram of the Distributed Memory Generator (8.0) is shown. It has four inputs: "a[5:0]", "d[31:0]", "clk", and "we". It has one output: "spo[31:0]".

Page Number: 15/16

仿真



Name	Value
clk	1
writeEn	1
addr[5:0]	00
Data[31:0]	12345678
ReadData[31:0]	12345678



```
1  `timescale 1ns / 1ps
2  module dRAM_Sim( );
3      logic      clk,   writeEn;
4      logic [5:0] addr;
5      logic [31:0] Data;
6      logic [31:0] ReadData;
7
8      dRAM d0 (
9          .a(addr),           // input wire [5 : 0] a
10         .d(Data),           // input wire [31 : 0] d
11         .clk(clk),          // input wire clk
12         .we(writeEn),       // input wire we
13         .spo(ReadData)      // output wire [31 : 0] spo
14     );
15
16     initial begin
17         clk = 0; writeEn = 0; addr = 0;
18         #60 Data = 32'h12345678;
19         #10 writeEn = 1;
20     end
21
22     always #10 clk = ~clk;
23 endmodule
```